SAS

STATISTICAL ANALYSIS SYSTEM
Parte II: procedure per l’analisi statistica dei dati

Introduzione al Proc Step di analisi

· Comincia sempre con la parola chiave PROC;

· Si compone di un insieme di istruzioni;

· Produce un output di default sulla finestra di output;

· Può produrre un output più complesso o più leggero utilizzando le opzioni;

· Attraverso alcune procedure di analisi si producono le misure più note per analisi statistiche descrittive (media, moda, mediana, misure di variabilità, misure di associazione, ecc.); alcune di queste sono prodotte dall’output di default altre possono essere richieste attraverso specifiche opzioni.

Sintassi di una PROC generica

PROC nome < options > ;

eventuali altre istruzioni < options >;

run;

Analisi dei dati: procedure preliminari

· operazioni di spoglio e controllo dei dati

· operazioni di ponderazioni dei dati

Operazioni di spoglio e controllo dei dati:

la Proc FREQ per analisi preliminari

L’istruzione PROC FREQ, è l’unica istruzione obbligatoria per utilizzare la procedura; se si specificano le istruzioni che seguono, la procedura da luogo a una tabella di frequenza semplice, per ogni variabile presente nel data set creato più di recente:

   proc freq;

   run;

Se si vuole analizzare uno specifico data set, è necessario utilizzare l’opzione data dell’istruzione proc freq:

   proc freq data=library.pippo;

   run;

Se invece si desidera analizzare solo alcune variabili è necessario utilizzare l’istruzione tables, come segue:

   proc freq;

   tables var1 var2 ….;
   run;

Sintassi:
	PROC FREQ <option(s)>; 

	TABLES request(s) </ option(s)>; 

	OUTPUT statistic-keyword(s) <OUT=SAS-data-set>; 

	TEST statistic-keyword(s); 

	FORMAT; 

	WEIGHT variable; 

	RUN; 

Options TABLES: 

􀂃NOFREQ 

􀂃NOPERCENT 

􀂃NOCUM 

􀂃NOROW 

􀂃NOCOL 

􀂃MISSING 


Esempio

L’esempio che segue riporta il codice della proc freq eseguita su 3 variabili del data set sui consumi delle famiglie:

data prova;set istat_md.consumi2;

proc freq data=prova;

tables regione numcomp sesso1;

run;

Segue l’output di default per la prima variabile analizzata, con le seguenti caratteristiche:

· nome della variabile e modalità;

· frequenze assolute;

· frequenze percentuali;

· frequenze cumulate assolute;

· frequenze cumulate percentuali;

· segnalazione dati mancanti “missing”, non conteggiati tra le frequenze

                                       The FREQ Procedure

                                             regione

                                                      Cumulative    Cumulative

                  regione    Frequency     Percent     Frequency      Percent

                  ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

                        1        2072        9.26          2072         9.26

                        3        2468       11.03          4540        20.30

                        4        1104        4.94          5644        25.23

                        5        1513        6.76          7157        32.00

                        6         643        2.87          7800        34.87

                        7         806        3.60          8606        38.47

                        8        1239        5.54          9845        44.01

                        9        1284        5.74         11129        49.75

                       10         610        2.73         11739        52.48

                       11         713        3.19         12452        55.67

                       12        1707        7.63         14159        63.30

                       13         811        3.63         14970        66.93

                       14         512        2.29         15482        69.21

                       15        1721        7.69         17203        76.91

                       16        1430        6.39         18633        83.30

                       17         430        1.92         19063        85.22

                       18         911        4.07         19974        89.30

                       19        1675        7.49         21649        96.79

                       20         719        3.21         22368       100.00

                                    Frequency Missing = 1550
ESEMPIO:

/*operazioni preliminary sul dataset*/

proc contents data=istat_md.consumi2 varnum; run;

********************************;
data new_var ; set  istat_md.consumi2;

if eta1 lt 25 then age_class=1;

if (25 le eta1 le 34) then age_class=2;

if (35 le eta1 le 44) then age_class=3;

if (45 le eta1 le 54) then age_class=4;

if (55 le eta1 le 64) then age_class=5;

if (eta1 ge 65) then age_class=6;

run;

**************************;
proc format;

value cond_prof  
1='occupato'
2='disoccupato'
3='in cerca prima occupazione'
4='casalinga'
5='studente'
6='inabile al lavoro'
7='pensionato/a'

8='in servizio di leva'

9='altra condizione' ;

run;

***********************; 

data a; set new_var; run;

proc freq data=a; tables conprof1; run;

proc freq data=a; tables conprof1 age_class; run;

proc freq data=new_var; tables conprof1;
format conprof1 cond_prof.; run;

Operazioni di ponderazione dei dati

Nella gestione di dati provenienti da indagini campionarie è pratica comune ricorrere a operazioni di ponderazione per supplire alle distorsioni derivanti dalla natura campionaria dei dati e/o per bilanciare i risultati in caso di missing.

Per effettuare questa operazione è necessario conoscere alcune variabili di struttura dell’universo di riferimento (es. genere, struttura per età, provenienza geografica, …).

Sulla base di queste conoscenze per ogni osservazione si produce un peso; in questo modo il campione “pesato” ricalca le caratteristiche dell’universo di riferimento.

Esempio

Come si può vedere nel riquadro, la struttura per genere dei rispondenti all’indagine sui consumi delle famiglie, mette in evidenza una prevalenza di capofamiglia maschi, pari al 75,23% del totale.

                                             sesso1

                                                     Cumulative    Cumulative

                  sesso1    Frequency     Percent     Frequency      Percent

                  ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

                       1       17669       75.23         17669        75.23

                       2        5819       24.77         23488       100.00

                                    Frequency Missing = 430

Da fonti esterne (per esempio dati Istat sul censimento 2001) si può ricavare la struttura per genere del capofamiglia, per l’universo delle famiglie italiane. Nella tabella che segue sono riportati i suddetti valori, si notino le differenze con la struttura per genere del campione.

	Genere
	%

	M
	72,321214

	F
	27,678786


Rapportando le due percentuali si ottiene il peso da attribuire ad ogni osservazione, secondo il genere:

Con delle semplici istruzioni di assegnazione si definiscono i pesi per ogni osservazione. 
Il programma che segue prevede anche l’esclusione dei dati mancanti:

data prova;set prova;

where sesso1 NE .;

if sesso1=1 then peso=0.961390;

else if sesso1=2 then peso=1.117235;

run;

Per effettuare la pesatura di ogni osservazione, nel calcolo delle frequenze, è sufficiente utilizzare l’istruzione weight della PROC FREQ

proc freq data=prova;

tables sesso1;

weight peso;

run;
Sulla finestra di output corrispondente si può notare la differenza nel conteggio delle frequenze e il conseguente allineamento della struttura del campione a quella della popolazione di riferimento secondo il genere.

                                             sesso1

                                                     Cumulative    Cumulative

                  sesso1    Frequency     Percent     Frequency      Percent

                  ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

                       1     16986.8       72.32       16986.8        72.32

                       2     6501.19       27.68      23487.99       100.00

N.B. NON CI SONO PIU’ I MISSING!!!!!!

Esercizio

Provare a realizzare la medesima operazione sullo stesso data set, considerando due variabili di struttura contemporaneamente: il genere e la provenienza geografica.

La struttura dell’universo di riferimento secondo le due variabili è riportata nella tabella seguente (valori percentuali):

[image: image2.png]Genere

Macroregione |Maschio Femmina
Nord 33,69099265 14,83000246
Centro 15,95739189 5,722685322

Sud 22,71991506 7,079012613





Nel data set sui consumi le variabili sono denominate sesso1 (1=M; 2=F) e macroregione (1=Nord; 2=Centro; 3=Sud).

data prova;set istat_md.consumi2;
proc freq;

tables sesso1*macroreg/out=sexmacr;

run;

/*pesi

mu_N_m=33.69099265/31.95=1.0544911627

mu_N_f=14.83000246/12.05=1.2307055983

mu_C_m=15.95739189/19.12=0.83459162604

mu_C_f=5.722685322/5.97=0.95857375577

mu_S_m=22.71991506/24.49=0.92772213393

mu_S_f=7.079012613/6.42=1.102649939719*/
data prova;set prova;

where (sesso1 NE .) and (macroreg ne .);

if sesso1=1 and macroreg=1 then peso=1.0544911627;

if sesso1=2 and macroreg=1 then peso=1.2307055983;

if sesso1=1 and macroreg=2 then peso=0.83459162604;

if sesso1=2 and macroreg=2 then peso=0.95857375577;

if sesso1=1 and macroreg=3 then peso=0.92772213393;

if sesso1=2 and macroreg=3 then peso=1.102649939719; run;
proc freq;

tables sesso1*macroreg/out=sexmacr;

weight peso;

run;
Analisi dei dati: procedure fondamentali

· Tabelle di frequenza e di contingenza

· Sintesi delle distribuzioni semplici e multivariate

Tabelle di frequenza e di contingenza

Istruzioni base

Come già introdotto nella sezione precedente, la procedura PROC FREQ consente di produrre tabelle di frequenza semplici, ma anche tabelle a doppia entrata, a tre variabili e così via fino a tabelle a n-entrate. Lo schema che segue riassume la sintassi di base della procedura.

	PROC FREQ <option(s)>; 

	
	
	

	
	TABLES request(s) </ option(s)>;
	

	
	OUTPUT statistic-keyword(s) <OUT=SAS-data-set>;
	
	

	
	TEST statistic-keyword(s); 
	
	
	

	
	FORMAT;
	
	
	
	

	
	WEIGHT variable;
	
	
	
	

	RUN;
	
	
	
	


Con l’istruzione TABLES si selezionano la\le variabili per le tabelle di frequenza semplici o multiple, nel secondo caso le variabili coinvolte sono incrociate con l’operatore “*”.

Esempio

Dopo aver attribuito etichette e formati, produrre una tabella semplice per il genere, l’area di provenienza e la numerosità familiare, e una a doppia entrata per il genere e l’area di provenienza.

/*ASSEGNAZIONE ETICHETTE ALLE VARIABILI*/
data prova; set istat_md.consumi2;

label sesso1='genere'

regione='macroregione'

numcomp='numerosità familiare';run;

/*ATTRIBUZIONE DEI FORMATI*/
proc format;

value sexfmt 1='M' 2='F';

value macrfmt 1-8='Nord' 9-12='Centro' 13-20='Sud+Isole';

value ncompfmt 1='1' 2='2' 3='3' 4='4' 5-10='5 o più';run;

/*ISTRUZIONI PER TABELLE*/
proc freq data=prova;

tables sesso1 regione numcomp sesso1*regione;

format sesso1 sexfmt. regione macrfmt. numcomp ncompfmt.;run;

Alcune opzioni dell’istruzione TABLES:

· NOFREQ (l’output non riporta le frequenze assolute)

· NOPERCENT (l’output non riporta le frequenze %)

· NOCUM (l’output non riporta le frequenze cumulate)

· NOROW (l’output non riporta le frequenze % su tot di riga)

· NOCOL (l’output non riporta le frequenze % su tot di colonna)

Esercizio

Sperimentare le opzioni sul codice riportato nel riquadro, e osservare le differenze sull’output.

Trattamento dei casi mancanti

La PROC FREQ non conteggia le frequenze dei missing di default; attraverso alcune opzioni dell’istruzione TABLES è possibile:

· stampare i missing sull’output senza che sia conteggiato nelle percentuali – opzione missprint;

· stampare i missing sull’output come se costituissero una modalità a tutti gli effetti – opzione missing;

/*ISTRUZIONI PER TABELLE DI FREQUENZA

stampa dei dati mancanti sull’output senza conteggiarli nelle percentuali*/
proc freq;

tables regione/missprint;

format regione macrfmt.;

run;

/*ISTRUZIONI PER TABELLE DI FREQUENZA

stampa dei dati mancanti sull’output, ritenuti validi a tutti gli effetti*/
proc freq;

tables regione/missing;

format regione macrfmt.;

run;

ESEMPIO:

proc freq;

tables numcomp*regione regione*sesso1;

format sesso1 sexfmt. regione macrfmt. numcomp ncompfmt.;

run;

proc freq;

tables numcomp*regione/norow;*provare ad aggiungere nopercent;
*tables regione*sesso1/nocol;

format sesso1 sexfmt. regione macrfmt. numcomp ncompfmt.;

run;

Calcolo dell’indice Chi-Quadrato

Attraverso l’opzione chisq dell’istruzione TABLES, si possono calcolare diverse misure di associazione per tabelle bivariate sulla base del più noto indice chi-quadrato.

Altri indici possono essere prodotti con l’opzione measures e con l’opzione all, che produce tutte le misure disponibili.

Esempio

Il codice nel riquadro che segue produce una tabella di contingenza incrociando il genere del capofamiglia con l’area di residenza. Le misure che seguono dipendono dall’opzione chisq, e riguardano il grado di associazione delle due variabili.

/*ISTRUZIONI PER TABELLE con calcolo indici CHI-QUADRATO*/
proc freq;

tables sesso1*regione/norow chisq;

format sesso1 sexfmt. regione macrfmt.;

run;

/*ISTRUZIONI PER TABELLE chi quadrato*/
proc freq;

tables numcomp*regione/norow chisq;

run;

Nel riquadro che segue è riportata la parte aggiuntiva di output, stampata in seguito all’opzione chisq. Alle prime tre statistiche è associata la relativa probabilità del test chi-quadro. Seguono gli indici di associazione Phi, coefficiente di contingenza e il V di Cramer.

                      Statistiche per la tabella di sesso1 per regione

                   Statistica                    DF      Valore      Prob

                   ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

                   Chi-quadrato                   2    113.9961    <.0001

                   Chi-quadrato rapp verosim      2    115.5174    <.0001

                   Chi-quadrato MH                1    103.9033    <.0001

                   Coefficiente Phi                      0.0720

                   Coefficiente di contingenza           0.0718

                   V di Cramer                           0.0720

                          Dimensione effettiva del campione = 22005

                            Frequenza dei valori mancanti = 1913

Creazione di un file SAS di output

· Attraverso l’opzione out dell’istruzione table, si possono salvare le frequenze (assolute e percentuali) di output su un file tipo data set.

proc freq data=prova;

tables sesso1*regione/out=tabella;

format sesso1 sexfmt. regione macrfmt. numcomp ncompfmt.;

run;
· Anche attraverso l’istruzione output è possibile, specificando le statistiche desiderate, produrre un data-set di output che le contiene.

proc freq data=prova;

tables sesso1* regione/chisq;

output chisq out=risult1 ;

format sesso1 sexfmt. regione macrfmt. numcomp ncompfmt.;

run;

In entrambi i casi può rivelarsi utile non stampare l’output nella relativa finestra; per ottenere ciò è sufficiente aggiungere l’opzione noprint all’istruzione proc freq.

Esercizio

Provare a realizzare una tabella di frequenza che incroci le tre variabili: genere, regione e numero di componenti della famiglia.

Sintesi delle distribuzioni semplici e multivariate

In questa sezione si analizzano le principali procedure che producono sintesi delle distribuzioni:

· PROC MEANS

· PROC UNIVARIATE

PROC MEANS

La Proc Means fornisce gli strumenti di sintesi per computare statistiche descrittive per variabili numeriche; le elaborazioni possono essere compiute su tutte le osservazioni oppure per gruppi di osservazioni omogenee classificate rispetto a una o più variabili. Per esempio la Proc Means:

· calcola statistiche descrittive basate sui momenti;

· calcola la mediana e tutti gli altri quantili;

· identifica i valori estremi;

· calcola l’intervallo di confidenza per la media.

	PROC MEANS <option(s)> <statistic-keyword(s)>; 

BY <DESCENDING> variable-1 <... <DESCENDING> variable-n><NOTSORTED>; 

CLASS variable(s) </ option(s)>; 

OUTPUT <OUT=SAS-data-set> <output-statistic-specification(s)> 
<id-group-specification(s)> <maximum-id-specification(s)> 
<minimum-id-specification(s)> </ option(s)> ; 

VAR variable(s) < / WEIGHT=weight-variable>; 

WEIGHT variable; 

RUN;




Procedura base

Il codice riportato nel riquadro mostra una procedura di tipo Means di base, applicata alle variabili relative alla spesa per elettricità e all’uso del telefono, presenti nel data set sui consumi delle famiglie.

data prova;set istat_md.consumi;

label c_elettr='consumi elettricità' c_tel='consumi telefonici';

proc means;

var c_elettr c_tel;

run;

Nel riquadro successivo è riportato l’output stampato nella relativa finestra.

                                      The MEANS Procedure

  Variable  Label                    N          Mean       Std Dev       Minimum       Maximum

  ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

  c_elettr  consumi elettricità  23877      65440.41      51613.88       6000.00     750000.00

  c_tel     consumi telefonici   22150      75583.69      50734.91      10000.00    1000000.00

  ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ
Commento all’output:
Notare che la terza colonna N riporta la numerosità delle osservazioni sulle quali sono state calcolate le misure. In entrambi i casi non si ritrova la numerosità totale del campione, pari a 23918 unità; questo perché la procedura effettua le elaborazioni escludendo i dati mancanti “missing”.

Ponderazione delle osservazioni: istruzione weight
Al programma precedente è stata aggiunta l’istruzione weight seguita dal nome della variabile di ponderazione. Si noti anche l’utilizzo dell’opzione data dell’istruzione proc means.

proc means data=prova;

var c_elettr c_tel;

weight w_gen;

run;

Si confronti ora il risultato di output riportato nel riquadro con quello precedente.

Notare:

· differenza nella numerosità delle osservazioni N

· differenza nelle medie e SQM

  Variable  Label                    N          Mean       Std Dev       Minimum       Maximum

  ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

  c_elettr  consumi elettricità  23452      64870.82      51135.09       6000.00     750000.00

  c_tel     consumi telefonici   21768      75234.45      50550.01      10000.00    1000000.00

  ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ
Istruzione proc means: statistiche opzionali

Le statistiche di default della procedura, come si può vedere negli esempi di output, sono la media semplice, la deviazione standard, il valore minimo e il valore massimo. Altre statistiche possono essere richieste opzionalmente utilizzando specifiche parole-chiave. La tabella seguente riporta una selezione delle parole chiave per le principali statistiche descrittive:

	Statistiche descrittive
	Indici di posizione

	Parola-chiave
	Indice
	Parola-chiave
	Indice

	
	
	
	

	CSS
	Devianza
	MEDIAN|P50
	Mediana 

	CV
	Coeff. variazione
	P1
	1° percentile

	KURTOSIS|KURT
	Indice di Curtosi
	P5
	5° percentile

	MAX
	Valore massimo
	P10
	10° percentile

	MEAN
	Media aritmetica
	Q1|P25
	1° quartile

	MIN
	Valore minimo
	Q3|P75
	3° quartile

	N
	Num. osser. valide
	P90
	90° percentile

	NMISS
	Num. osser.manc.
	P95
	95° percentile

	RANGE
	Interv. di variaz.
	P99
	99° percentile

	SKEWNESS|SKEW
	Indice di simmetr.
	QRANGE
	Dist. interquartile

	STDDEV|STD
	Dev. Standard
	
	

	SUM
	Somma dei valori
	
	

	SUMWGT
	Somma dei pesi
	
	

	USS
	Somma val. quadr.
	
	

	VAR
	Varianza
	
	


Esempio

Si vogliono analizzare ancora una volta la spesa per i consumi di elettricità e per quelli telefonici. Stavolta però si vuole produrre un output che contenga il numero di osservazioni valide, la somma dei valori di spesa individuali, la media aritmetica e il coefficiente di variazione.

I riquadri che seguono riportano il codice predisposto e il relativo output.

proc means data=prova N sum mean cv;

var c_elettr c_tel;

run;

                                      The MEANS Procedure

                                                                                    Coeff of

    Variable    Label                      N             Sum            Mean       Variation

    ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

    c_elettr    consumi elettricità    23877      1562520749        65440.41      78.8715644

    c_tel       consumi telefonici     22150      1674178756        75583.69      67.1241510

    ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ
Analisi per gruppi di osservazioni omogenee

i. Istruzione BY

Produce le statistiche relative alla variabile d’analisi, separatamente per ogni modalità della variabile di classificazione (una o più) specificata con l’istruzione BY.

Per eseguire l’istruzione BY è necessario, prima di eseguire la proc, ordinare il data set in maniera crescente secondo la variabile di classificazione.

Segue un programma esemplificativo e l’output relativo.

proc sort data=prova;

by sesso1;

proc means data=prova N sum mean cv;

var c_elettr c_tel;

by sesso1;

format sesso1 sexfmt.;

run;
------------------------------------------- genere=. -------------------------------------------

                                                                                    Coeff of

    Variable    Label                      N             Sum            Mean       Variation

    ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

    c_elettr    consumi elettricità      425     28239582.00        66446.08     101.3807219

    c_tel       consumi telefonici       382     28674750.00        75064.79      66.7099223

    ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

------------------------------------------- genere=M -------------------------------------------

                                                                                    Coeff of

    Variable    Label                      N             Sum            Mean       Variation

    ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

    c_elettr    consumi elettricità    17644      1237280015        70124.69      74.6000230

    c_tel       consumi telefonici     16603      1305037762        78602.53      66.0839965

    ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

------------------------------------------- genere=F -------------------------------------------

                                                                                    Coeff of

    Variable    Label                      N             Sum            Mean       Variation

    ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

    c_elettr    consumi elettricità     5808       297001152        51136.56      88.2372498

    c_tel       consumi telefonici      5165       340466244        65917.96      68.8230171

    ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

ii. Istruzione CLASS

Le statistiche relative a gruppi di osservazioni omogenee, originate da una variabile di classificazione (una o più), sono prodotte con un’unica tavola di output.

Segue un programma esemplificativo e l’output relativo.

proc means data=prova1 N sum mean cv;

var c_elettr c_tel;

class sesso1;

format sesso1 sexfmt.;

run;

                                      The MEANS Procedure

                                                                                        Coeff of

sesso1    N Obs    Variable    Label           N             Sum            Mean       Variation

ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

M         17669    c_elettr    c_elettr    17644      1237280015        70124.69      74.6000230

                   c_tel       c_tel       16603      1305037762        78602.53      66.0839965

F          5819    c_elettr    c_elettr     5808       297001152        51136.56      88.2372498

                   c_tel       c_tel        5165       340466244        65917.96      68.8230171

ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ
iii. Più variabili di classificazione

Le istruzioni by e class possono essere applicate con più variabili di classificazione contemporaneamente; i nomi delle variabili devono seguire l’istruzione e devono essere separate da spazi.

Esercizio

Eseguire una proc means per la spesa relativa ai consumi da gas di rete (variabile “c_gas_r”), presente nel data set sui consumi delle famiglie, utilizzando alternativamente le istruzioni class e by con le variabili regione e numero di componenti della famiglia. Oltre alla numerosità di osservazioni e alla media aritmetica, richiedere la mediana, il primo e il terzo quartile, e il numero di casi mancanti.

Creazione di un file di output

Le statistiche prodotte dalla proc means possono essere opportunamente denominate e salvate su un apposito file data set.

Il seguente programma effettua le suddette elaborazioni per le variabili relative alle spese per consumo di gas e di acqua.



nomi delle variabili del file di output

proc means data=prova N mean median cv noprint;

var c_gas_r c_acqua;

output out=indici N=obs_gas obs_acq mean=m_gas m_acq

median=md_gas md_acq cv=cv_gas cv_acq;

run;

Trattamento dei dati mancanti

Come già osservato in precedenza, il calcolo della media e delle altre statistiche descrittive non può che escludere i missing dalla computazione.

Nel caso di analisi per gruppi di osservazioni omogenee può accadere che la variabile di analisi sia missing, ma non lo sia quella di classificazione.

In questo caso l’utilizzo dell’istruzione BY mette in evidenza queste osservazioni producendo una modalità missing distinta dal simbolo “.”. Se invece si adopera l’istruzione CLASS, le osservazioni missing per la variabile di classificazione sono esclusi dal calcolo della media; per includerle nella suddetta analisi è sufficiente utilizzare l’opzione missing nell’istruzione proc means.

PROC UNIVARIATE

La Proc Univariate produce statistiche descrittive semplici per variabili numeriche. Per esempio la Proc univariate:

· calcola statistiche descrittive basate sui momenti;

· calcola la mediana, la moda, il range e i quantili;

· crea il plot della distribuzione di una variabile;

· realizza plot di normalità;

· crea data set di output contenente gli indici richiesti.

Sintassi di base:

	PROC UNIVARIATE <option(s)>; 

BY <DESCENDING> variable-1 <...<DESCENDING> variable-n><NOTSORTED>; 

CLASS variable-1<(variable-option(s))> <variable-2<(variable-option(s))>> 
</ KEYLEVEL='value1'|('value1' 'value2')>; 

OUTPUT <OUT=SAS-data-set>statistic-keyword-1=name(s)<... statistic-keyword-n=name(s)> <percentiles-specification>; 

ID variable;
VAR variable(s); 

WEIGHT variable; 

RUN;




Procedura base

Il codice riportato nel riquadro mostra una procedura di tipo Univariate di base, applicata alla variabili relative alla spesa per acqua, presente nel data set sui consumi delle famiglie.

proc univariate data=prova;

var c_acqua;

run;

Nei riquadri che seguono sono state riportate le due pagine di output che la procedura produce di default.

                                   La procedura UNIVARIATE

                               Variabile:  c_acqua  (c_acqua)

                                           Momenti

           N                       14569    Somma dei pesi                   14569

           Media              35606.5039    Somma delle osservazioni     518751155

           Deviazione std     33261.5719    Varianza                    1106332168

           Skewness           2.95468355    Kurtosis                    13.5039107

           SS non corretta     3.4588E13    SS corretta                  1.6117E13

           Coeff variaz       93.4143157    Errore std media            275.567438

                                 Misure statistiche di base

                    Posizione                       Variabilità

                Media     35606.50     Deviazione std                  33262

                Mediana   25067.00     Varianza                   1106332168

                Moda      16667.00     Intervallo                     330000

                                       Intervallo interquartile        28333

                                  Test di posizione: Mu0=0

                 Test                    -Statistica-    -----Valore p-----

                 T di Student            t   129.2116    Pr > |t|    <.0001

                 dei segni               M     7284.5    Pr >= |M|   <.0001

                 dei segni per ranghi    S   53067583    Pr >= |S|   <.0001

                                  Quantili (Definizione 5)

                                   Quantile         Stima

                                   100% Max        333333

                                   99%             166667

                                   95%             100000

                                   90%              70667

                                   75% Q3           43333

                                   50% Mediana      25067

                                   25% Q1           15000

                                   10%              10000

                                   5%                7333

                                   1%                4667

                                   0% Min            3333

Notare che sulla parte dedicata al test di locazione, sono riportati i risultati di un confronto tra la media della distribuzione e una media pari a 0, attraverso il test t. Si provi a inserire un altro valore medio di confronto mediante l’opzione dell’istruzione principale mu0=…
                                   La procedura UNIVARIATE

                               Variabile:  c_acqua  (c_acqua)

                                     Osservazioni estreme

                           ---Inferiori---        ----Superiori---

                           Valore      Oss         Valore      Oss

                             3333    23038         333333     5732

                             3333    22279         333333     8794

                             3333    21910         333333    13862

                             3333    21757         333333    20399

                             3333    21665         333333    21410

                                       Valori mancanti

                                                 ---Percentuale di---

                          Valore                    Tutte         Oss

                        mancante    Conteggio      le oss    mancanti

                               .         9349       39.09      100.00
Contenuto dell’output

All’interno dell’output di default si possono distinguere 6 gruppi di statistiche (alcune ricorrono in più di un gruppo): 

a) moments, statistiche basate sui momenti;

b) misure statistiche di base;

c) test di locazione ((0=0 sotto H0);

d) quantili;

e) valori estremi;

f) missing.

Come si può notare le statistiche computate sono varie e numerose, l’output stampato è quindi molto ingombrante. Se si desidera un output più compatto è consigliabile dare l’opzione noprint (all’istruzione iniziale) e creare un data set di output con le statistiche desiderate (vedere più avanti la sezione Produzione di data-set di output).

Opzione plot

L’istruzione principale, seguita dall’opzione plot, produce, oltre all’output di default esaminato nella sezione precedente, alcuni grafici relativi alla distribuzione della variabile di analisi. 

Il primo grafico si compone da un istogramma di frequenza (stem-leaf) e un box-plot; il secondo è un normal probability plot.

Seguono il programma contenente l’opzione e i grafici estratti dall’output relativo.

proc univariate data=prova plot;

var eta1;

run;

                                    The UNIVARIATE Procedure

                                    Variable:  eta1  (eta1)

                                            Moments

                N                       23751    Sum Weights              23751

                Mean               55.4235611    Sum Observations       1316365

(...)

                                 Histogram                          #             Boxplot

         102.5+*                                                    5                |

              .*                                                   29                |

              .***                                                156                |

              .*********                                          472                |

              .*****************                                  859                |

              .******************************                    1566                |

              .****************************************          2076                |

              .*******************************************       2227             +-----+

              .********************************************      2307             |     |

              .*******************************************       2245             *--+--*

              .************************************************  2521             |     |

              .************************************************  2515             |     |

              .*********************************************     2349             +-----+

              .********************************************      2292                |

              .*****************************                     1505                |

              .**********                                         529                |

              .**                                                  95                |

          17.5+*                                                    3                |

               ----+----+----+----+----+----+----+----+----+---

               * may represent up to 53 counts

                                    The UNIVARIATE Procedure

                                    Variable:  eta1  (eta1)

                                         Normal Probability Plot

                     102.5+                                                  *

                          |                                                  *

                          |                                                +**

                          |                                            ******

                          |                                        *****

                          |                                    *****

                          |                                 ****+

                          |                              ****+

                          |                           ****+

                          |                         ***

                          |                      ****

                          |                   ****

                          |                ****

                          |            *****

                          |      *******

                          |*******++

                          |*  +++

                      17.5+*++

                           +----+----+----+----+----+----+----+----+----+----+

                               -2        -1         0        +1        +2

Istruzioni BY e CLASS

Nella Proc Univariate le istruzioni BY e CLASS, producono degli output molto simili (in misura ancora maggiore di quanto riscontrato per la Proc Means) ; anche in questo caso con l’istruzione BY la Proc Univariate deve essere preceduta da quella Sort, secondo la medesima variabile secondo cui si intende classificare. Per ogni modalità della variabile di classificazione vengono stampate le consuete due pagine di output, con tutte le statistiche prodotte dalla procedura Univariate. La differenza fondamentale è data dall’analisi per la modalità missing, che è presente solo nell’output dell’istruzione BY.

Se entrambe le istruzioni sono applicate insieme all’opzione plot nell’istruzione iniziale Proc Univariate l’ulteriore differenza è costituita dal confronto dei grafici box-plot; infatti, utilizzando l’istruzione BY, i grafici sono stampati affiancati su un’unica tavola di output, uno per ogni modalità della variabile di classificazione.

Esercizio

Sperimentare le differenze originate dalle istruzioni BY e CLASS, utilizzando come variabile di analisi una variabile quantitativa a scelta tra quelle sulle spese domestiche presenti nel data set sui consumi delle famiglie e come variabile di classificazione, una a scelta tra: genere, regione, numero di componenti della famiglia, numero di stanze (opportunamente formattate se necessario!).

Produzione di data set di output

Le istruzioni per produrre un data set di output con la Proc Univariate, risultano simili a quelle già viste in precedenza con la Proc Means (pag. 51). Anche le parole chiave da utilizzare per selezionare gli indici da inserire nel file di output sono le stesse esaminate per la precedente procedura (per visualizzare ulteriori indici e relative parole chiave si consulti l’help del programma SAS).

Segue il programma completo di istruzione output e la stampa della finestra di output.

proc univariate data=prova noprint;

var eta1;

class sesso1;

output out=indici mode=modeta  q1=q1eta median=medeta



q3=q3eta;

format sesso1 sexfmt.;

run;

proc print;

run;

                      Obs    sesso1    q3eta    medeta    q1eta    modeta

                       1       M         64       52        42       55

                       2       F         75       65        48       71
Si noti che l’output in finestra è stato stampato solo perché il programma termina con l’istruzione proc print; in caso contrario l’opzione noprint, presente nell’istruzione principale, agisce sopprimendo la stampa dell’output nella relativa finestra.

Produzione di data set di output per il calcolo dei percentili

Come già accennato in precedenza, la procedura Univariate può produrre una vasta gamma di statistiche opzionali (si veda ancora l’elenco delle parole chiave a pagina 51) attraverso l’istruzione output, contenute ovviamente in un file di output. I percentili disponibili attraverso parole chiave sono il 1°, 5°, 10°, 90°, 95° e il 99° percentile. Se si desiderano degli altri percentili bisogna opportunamente programmare l’istruzione output attraverso le seguenti opzioni:

pctlpts= indica quali percentili devono essere calcolati (esempio il 20° e il 40°);

pctlpre= indica il prefisso da utilizzare per la denominazione dei percentili nel file di output;

pctlname= indica il nome da utilizzare per la denominazione dei percentili nel file di output.

Segue il programma completo di istruzione output e la stampa della finestra di output.

proc univariate data=prova noprint; 

      var eta1 c_elettr; 

      output out=Perc pctlpts=20 40 pctlpre=eta elettr

      pctlname=P20 P40;run;

proc print;

   run;
                                   eta    eta    elettr    elettr

                            Oss    P20    P40      P20       P40

                             1      40     50     30000     42500

ESEMPIO PLOT:

proc univariate data=istat_md.consumi2 plot;


var eta1;

  run;

proc univariate data=istat_md.consumi2; 

      var eta1; 

      histogram / cframe = ligr

                  cfill  = blue





  normal(color=red );  

   run;

proc univariate data=istat_md.consumi2; 

      var c_elettr; 

      histogram / normal




  lognormal (l=1  color=red)

                  weibull   (l=2  color=blue)

                  gamma     (l=8  color=yellow); 

      inset mean std normal; 

   run;

proc univariate data=istat_md.consumi2 noprint;

      class macroreg /*sesso1*/;

      histogram canmfig / nrows      = 3
                         cframe     = ligr

                         cfill      = red

                         cframeside = ligr;

   run;

proc univariate data=istat_md.consumi2  noprint;

      qqplot c_elettr /lognormal(sigma=est) cframe = ligr;

   run;

ods pdf file='C:\CORSO SAS\esempi lez .pdf';

 proc univariate data=istat_md.consumi2  noprint;

      qqplot c_elettr /lognormal(sigma=est) cframe = ligr;

   run;

ods  _all_  close;

ods listing;

Analisi dell’associazione tra variabili

· analisi preliminari

· studio dell’associazione

· variabile risposta categorica

Analisi preliminari: la PROC PLOT

Nella fase preliminare dello studio dell’associazione tra variabili quantitative può essere utile visualizzare il diagramma scatter. Questo strumento è fornito dalla procedura PLOT.

Sintassi di base

	PROC PLOT <option(s)>; 

BY <DESCENDING> variable-1 <...<DESCENDING> variable-n> 
<NOTSORTED>; 
PLOT plot-request(s) </ option(s)>; 
RUN;



Procedura di base

data prova;set istat_md.consumi2;

proc plot;

plot c_elettr*c_acqua;

run;

Il programma riportato nel riquadro produce un diagramma scatter tra la spesa per il consumo di elettricità e la spesa per il consumo di acqua. Le variabili si inseriscono con l’istruzione plot (per prima quella che si desidera sull’asse delle ordinate), separate da asterisco. Segue l’output nella pagina seguente.

Specificando nell’istruzione plot una qualche variabile di classificazione preceduta dal segno “=” i punti dello scatter saranno distinti secondo le modalità di suddetta variabile. PROVARE!
                 Plot of c_elettr*c_acqua.  Legend: A = 1 obs, B = 2 obs, etc.

NOTE: 9349 obs had missing values.  9722 obs hidden.

Notare la legenda in alto: lettera maiuscola=numero di osservazioni!

data prova;set istat_md.consumi2;

proc plot;

plot c_elettr*c_acqua=sesso1;

format sesso1 sexfmt.;

run;

symbol1 v=dot c=blue i=rl ci=red;

proc gplot data=prova;

plot c_elettr*c_acqua;

run;
istruzione SYMBOL

V= definisce simbolo dei punti

C= definisce colore punti

I= definisce tipo di interpolazione (RL: lineare)

CI= definisce colore linea interpolante

Studio dell’associazione tra variabili:

· PROC CORR

· PROC REG

· PROC ANOVA

PROC CORR

La Proc Corr computa il coefficiente di correlazione di Pearson, tre misure di associazione non parametrica (opzionali) e le probabilità associate a queste statistiche.

Sintassi di base

PROC CORR <option(s)>; 

	
	BY <DESCENDING> variable-1<...<DESCENDING> variable-n> <NOTSORTED>; 

	
	PARTIAL variable(s); 

	
	VAR variable(s); 

	
	WEIGHT weight-variable; 

	RUN;
	


Procedura base

Nella procedura base è sufficiente, oltre all’istruzione obbligatoria proc corr, elencare le variabili (separate da spazi) sulle quali calcolare il coefficiente di correlazione lineare mediante l’istruzione var. Il programma che segue produce la matrice di correlazione per alcune variabili di spesa del data set sui consumi delle famiglie.

data prova;set istat_md.consumi2;run;

proc corr;

var c_elettr c_acqua c_gas_r numcomp eta1 stanze;

run;

Lettura dell’output

La prima parte dell’output (si veda il riquadro seguente) riporta le principali statistiche descrittive per ogni variabile. Segue la vera e propria matrice di correlazione sulla quale, per ogni incrocio, sono stampati: il coefficiente di correlazione di Pearson, la probabilità che r=0 sotto ipotesi nulla e il numero di osservazioni valide.

                                       The CORR Procedure

              6  Variables:    c_elettr c_acqua  c_gas_r  numcomp  eta1     stanze

                                       Simple Statistics

Variable          N         Mean      Std Dev          Sum      Minimum      Maximum   Label

c_elettr      23877        65440        51614   1562520749         6000       750000   c_elettr

c_acqua       14569        35607        33262    518751155         3333       333333   c_acqua

c_gas_r       15749       106400       110978   1675691786         5000      1000000   c_gas_r

numcomp       23918      2.72719      1.30323        65229      1.00000     10.00000   numcomp

eta1          23751     55.42356     15.58377      1316365     16.00000    101.00000   eta1

stanze        23877      4.40524      1.47062       105184      1.00000     20.00000   stanze

                               Pearson Correlation Coefficients

                                  Prob > |r| under H0: Rho=0

                                     Number of Observations

                c_elettr       c_acqua       c_gas_r       numcomp          eta1        stanze

  c_elettr       1.00000       0.21560       0.24634       0.34204      -0.10554       0.26246

  c_elettr                      <.0001        <.0001        <.0001        <.0001        <.0001

                   23877         14569         15749         23877         23710         23877

  c_acqua        0.21560       1.00000       0.14119       0.24319      -0.04299       0.15420

  c_acqua         <.0001                      <.0001        <.0001        <.0001        <.0001

                   14569         14569         10472         14569         14454         14569

  c_gas_r        0.24634       0.14119       1.00000       0.13848      -0.00203       0.25092

  c_gas_r         <.0001        <.0001                      <.0001        0.7996        <.0001

                   15749         10472         15749         15749         15656         15749

  numcomp        0.34204       0.24319       0.13848       1.00000      -0.33261       0.29005

  numcomp         <.0001        <.0001        <.0001                      <.0001        <.0001

                   23877         14569         15749         23918         23751         23877

  eta1          -0.10554      -0.04299      -0.00203      -0.33261       1.00000      -0.02447

  eta1            <.0001        <.0001        0.7996        <.0001                      0.0002

                   23710         14454         15656         23751         23751         23710

  stanze         0.26246       0.15420       0.25092       0.29005      -0.02447       1.00000

  stanze          <.0001        <.0001        <.0001        <.0001        0.0002

                   23877         14569         15749         23877         23710         23877
Principali opzioni dell’istruzione Proc Corr

Se si desidera sopprimere la prima parte dell’output è necessario utilizzare l’opzione nosimple nell’istruzione principale; se si desidera produrre una matrice di covarianza utilizzare l’opzione cov.

Il programma che segue applica entrambe le opzioni sopra esaminate.

proc corr data=prova nosimple cov;

var c_elettr c_acqua numcomp eta1 stanze;

run;

Se si desidera salvare la matrice di correlazione su un dataset di output utilizzare l’opzione outp=nome del file per l’istruzione principale proc corr.

data prova;set istat_md.consumi2;run;

proc corr /*outp=corr*/;

var c_elettr c_acqua numcomp eta1 stanze;

run;

proc corr data=prova nosimple cov pearson;

var c_elettr c_acqua numcomp eta1 stanze;

run;

PROC REG

La procedura REG è uno dei numerosi strumenti per l’analisi della regressione disponibili nel sistema SAS. La Proc Reg si propone obiettivi piuttosto generici, mentre altre procedure finalizzate alla regressione forniscono applicazioni più specialistiche.

Sintassi di base

PROC REG < options > ; 
< label: > MODEL dependents=<regressors> < / options > ; 

BY variables ; 

WEIGHT variable ; 

OUTPUT < OUT=SAS-data-set > keyword=names 

      < ... keyword=names > ; 

PLOT <yvariable*xvariable> <=symbol> 

      < ...yvariable*xvariable> <=symbol> < / options > ; 

PRINT < options > < ANOVA > < MODELDATA > ;
RUN;

Procedura base

Il programma che segue esegue una Proc Reg di base che analizza la regressione lineare tra la spesa per il consumo di elettricità e la numerosità familiare. 

proc reg data=prova;

label numcomp='numerosità familiare';

model c_elettr=numcomp;

run;

Opzionalmente può essere inserita l’istruzione plot che produce un diagramma scatter con retta di regressione e relativa equazione. Segue il programma completo di istruzione plot.

proc reg data=prova;

model c_elettr=numcomp;

plot c_elettr*numcomp;

run;

Lettura dell’output

Nel riquadro seguente è riportato l’output della Proc Reg di default. Come si può osservare la prima parte è dedicata all’analisi della varianza mentre nella seconda sono riportate le stime dei parametri con relativo test t.

La procedura REG

Modello: MODEL1

                     Variabile dipendente: c_elettr consumi elettricità

                   Numero di osservazioni lette                     23918

                   Numero di osservazioni usate                     23877

                   Numero di osservazioni con valori mancanti          41

                                   Analisi della varianza

                                        Somma dei      Media dei     Valore

       Origine                  DF       quadrati       quadrati          F    Pr > F

       Modello                   1    7.441194E12    7.441194E12    3163.19    <.0001

       Errore                23875    5.616429E13     2352430897

       Totale corretto       23876    6.360548E13

                 Radice MSE                48502    R-quadro         0.1170

                 Media dipendente          65440    R-quadr corr     0.1170

                 Coeff var              74.11607

                                     Stime dei parametri

                                             Stima dei         Errore

Variabile    Etichetta               DF      parametri       standard    Valore t    Pr > |t|

Interc       Interc                   1          28491      728.09347       39.13      <.0001

numcomp      numerosità familiare     1          13546      240.84890       56.24      <.0001

proc reg data=prova;

label c_elettr='consumi elettricità' c_tel='consumi telefonici';

model c_elettr=c_tel;

plot c_elettr*c_tel;

run;

ANALISI INTERETTIVA

data prova;set istat_md.consumi2;run;

proc reg lineprinter;

model c_acqua=c_elettr;

plot c_acqua*c_elettr;

run;

/* plot dei residui e dei valori previsti

  plot r.*p.; 

   run;*/
symbol1 c=blue; 

   proc reg data=prova; 

var c_tel;

model c_acqua=c_elettr /* r cli clm*/; 

      plot r.*p. / cframe=ligr; 

   run;

/*per terminare la reg:

   quit;

*/
   /*"Interactive Analysis"*/
/*posso aggiungere  variabili così se prima specificate nella var*/
   add c_tel; 

   print; 

   plot / cframe=ligr; 

   run;

/*costruisco grafico scatter con consumi acqua*elettrici, valori previsti c.acqua*elettrici

   ed intervallo di confidenza c-acqua*eletrici)*/
      symbol1 v=dot     c=yellow h=.3; 

   symbol2 v=square  c=red; 

   symbol3 f=simplex c=blue  h=2 v='-'; 

   symbol4 f=simplex c=blue  h=2 v='-'; 

   plot (c_acqua predicted. u95. l95.)*c_elettr 

        / overlay cframe=ligr; 

   run; 

/*BY*/
proc sort data=prova; by macroreg; run;

proc reg lineprinter;

model c_acqua=c_elettr;

by macroreg;

plot c_acqua*c_elettr;

run;

/*inserisco varibile dicotomica*/
data prova; set prova;

if numperc>=2 then perc2=1;

else perc2=0; run;

proc reg lineprinter;

model c_acqua=c_elettr perc2;

plot c_acqua*c_elettr;

run;

 output out=out r=r p=p; 

   run; 

   symbol1 v='0' c=blue   f=swissb; 

   symbol2 v='1' c=yellow f=swissb; 

   axis1 label=(angle=90); 

   proc gplot data=out; 

      plot r*p=perc2    / nolegend vaxis=axis1 cframe=ligr; 

      plot p*c_elettr=perc2 / nolegend vaxis=axis1 cframe=ligr; 

   run;

MODELLI NIDIFICATI

/*CREO LE VARIABILI DUMMY*/
data prova;set istat_md.consumi2;

if sesso1=1 then sesso1=1;

else if sesso1=2 then sesso1=0;

else sesso1=sesso1;

if p_lavast=1 then p_lavast=1;

else if p_lavast=2 then p_lavast=0;

else p_lavast=p_lavast;

if p_arcond=1 then p_arcond=1;

else if p_arcond=2 then p_arcond=0;

else p_arcond=p_arcond;

if p_pc=1 then p_pc=1;

else if p_pc=2 then p_pc=0;

else p_pc=p_pc;

if p_hifi=1 then p_hifi=1;

else if p_hifi=2 then p_hifi=0;

else p_hifi=p_hifi;

run;

/*STIMA DI MODELLI DI REGRESSIONE CON L'AGGIUNTA DI UNA VARIABILE AD OGNI NUOVO MODELLO*/
proc reg data=prova;

*model c_elettr=numcomp sesso1 ;
*model c_elettr=numcomp sesso1 eta1 ;
*model c_elettr=numcomp sesso1 eta1 stanze ;
*model c_elettr=numcomp sesso1 eta1 stanze p_lavast;
*model c_elettr=numcomp sesso1 eta1 stanze p_lavast p_arcond;
*model c_elettr=numcomp sesso1 eta1 stanze p_lavast p_arcond p_pc;
model c_elettr=numcomp sesso1 eta1 stanze p_lavast p_arcond p_pc p_hifi/CLB ALPHA=0.01;

run;

REG. BACKWARD/FORWARD/MAXR
proc reg data=prova; 

      model c_tel=macroreg numcomp numperc sesso1 titstu1 conprof1 

            /selection=maxr;

            /*   selection=forward; selection=backward;

   run;

PROC ANOVA

La procedura ANOVA è il principale strumento SAS per effettuare l’analisi della varianza; applica quindi una serie di metodi statistici atti a verificare le differenze fra le medie di una variabile dipendente in gruppi diversi, identificati dai valori delle variabili esplicative. 

Sintassi di base

PROC ANOVA < options >;
CLASS variables;
MODEL dependents=effects < / options >;
RUN;

Procedura di base

Il programma che segue analizza la variabile quantitativa consumi per elettricità rispetto ai gruppi generati dall’incrocio di due variabili categoriche: regione e numero di componenti, opportunamente specificate con l’istruzione class e come variabili esplicative nell’istruzione model.

proc anova data=prova;

class regione numcomp;

model c_elettr=regione numcomp;

format regione macrfmt. numcomp ncompfmt.;

run;

L’output prodotto ricalca la struttura della parte relativa all’analisi della varianza di quello della Proc Reg, naturalmente è da interpretare secondo il confronto tra le medie dei gruppi.

Attenzione: la Proc Anova, fornisce risultati validi se i dati provengono da un disegno bilanciato, ossia se per ogni sottogruppo è disponibile lo stesso numero di osservazioni n.

In caso contrario il risultato non è attendibile (così come nel nostro caso) e ciò viene segnalato sul log insieme alla proposta di utilizzo della procedura GLM (General Linear Models) come alternativa. Segue il programma che usa la procedura GLM.

proc glm data=prova;

class regione numcomp;

model c_elettr=regione numcomp;

format regione macrfmt. numcomp ncompfmt.;

run;
Utilizzando la procedura GLM viene prodotto il seguente output.

                                      La procedura GLM

                        Informazioni sui livelli di classificazione

                       Classe       Livelli    Valori

                       regione            3    Centro Nord Sud+Isole

                       numcomp            5    1 2 3 4 5 o più

                          Numero di osservazioni lette       23918

                          Numero di osservazioni usate       22330                                  

Variabile dipendente: c_elettr   consumi elettricità

                                          Somma dei           Media     Valore

     Origine                     DF        quadrati      quadratica          F    Pr > F

     Modello                      6    6.3884088E12    1.0647348E12     462.10    <.0001

     Errore                   22323    5.1435308E13    2304139602.9

     Totale corretto          22329    5.7823717E13

                   R-quadro     Var coeff    Radice MSE    c_elettr Media

                   0.110481      74.37551      48001.45          64539.33

                                                              Media     Valore

     Origine                     DF       Tipo I SS      quadratica          F    Pr > F

     regione                      2    152039068295     76019534148      32.99    <.0001

     numcomp                      4    6.2363697E12    1.5590924E12     676.65    <.0001

                                                              Media     Valore

     Origine                     DF     Tipo III SS      quadratica          F    Pr > F

     regione                      2    2708007235.6    1354003617.8       0.59    0.5556

     numcomp                      4    6.2363697E12    1.5590924E12     676.65    <.0001

Il test F generale risulta significativo, ad indicare una forte evidenza che le medie dei consumi di energia elettrica per le 15 (5x3) celle sono differenti. A questo punto si può analizzare questa differenza esaminando i test individuali per ogni effetto.

Le somme dei quadrati di tipo I e III non sono uguali quando, come nel nostro caso il disegno non è bilanciato; sempre per casi non bilanciati è preferibile utilizzare esclusivamente le somme di tipo III.

data pranova;set istat_md.consumi2;

where sesso1 ne . & statociv1 ne .;run;

proc anova;

class statociv1;

model sesso1=statociv1;

run;

data pranova1;set istat_md.consumi2;

where sesso1 ne . & statociv1 ne . & eta1 ne .;run;

proc anova;

class statociv1 eta1;

model sesso1=statociv1 eta1;

run;

proc anova data=prova;

class macroreg numcomp;

model c_elettr=macroreg numcomp;

*format macroreg macrfmt. numcomp ncompfmt.;
run;

Variabile risposta categorica:

PROC LOGISTIC

La Proc Logistic costituisce uno strumento per studiare la relazione tra una variabile risposta categorica e un insieme di variabili esplicative.

Sintassi di base

PROC LOGISTIC < options >; 
MODEL events/trials = < effects > < / options >; 

MODEL variable < (variable_options) > = < effects > < / options >; 

OUTPUT < OUT=SAS-data-set > 
                          < keyword=name...keyword=name > / < option >; 

< label: > TEST equation1 < , ... , < equationk >> < /option >; 

WEIGHT variable </ option >;
RUN;

Procedura di base

Nel programma che segue vengono create una serie di variabili dicotomiche che serviranno per l’implementazione di un modello di regressione logistica. A questo proposito è importante sottolineare che la variabile risposta (variabile dipendente) deve essere rigorosamente binaria (ancora meglio se 0-1, dummy) mentre le covariate (variabili indipendenti) possono essere qualitative dicotomiche (meglio se 0-1) o quantitative, ma mai qualitative politomiche.

Prima di creare delle variabili dicotomiche è opportuno verificare che le frequenze per modalità non siano inferiori al 10-15% in percentuale. Una variabile che assume lo stesso valore per quasi tutte le osservazioni spiegherebbe poco di un fenomeno!

Il programma che segue serve a creare delle variabili dummy, sulla base di alcune disponibili nel dataset istat_md.consumi2, secondo il seguente schema:

	Variabili di origine
	Variabili Dummy risultanti

	Clasred, classi di reddito, 14 categorie
	Topred=1 se la famiglia percepisce un redditomensile>=2000€ circa

	Numperc, numero dei percettori di reddito presenti in famiglia
	Perc2=1 se nella famiglia ci sono almeno 2 percettori di reddito

	Posprof1, posizione nella professione della persona di riferimento, 9 categorie
	Dir=1 se la persona di riferimento è dirigente o quadro

	P_…, possesso di alcuni beni durevoli, 2 categorie 1=sì 2=no
	D_…=1 se il bene è posseduto


data prova;set prova;

if clasred>=8 then topred=1;

else topred=0;

if numperc>=2 then perc2=1;

else perc2=0;

if (posprof1=1 or posprof1=2) then dir=1;

else dir=0;

if p_frigo=1 then d_frigo=1;

else d_frigo=0;

if p_lavast=1 then d_lavast=1;

else d_lavast=0;

if p_lavatr=1 then d_lavatr=1;

else d_lavatr=0;

if p_ap_pul=1 then d_ap_pul=1;

else d_ap_pul=0;

if p_arcond=1 then d_arcond=1;

else d_arcond=0;

if p_auto=1 then d_auto=1;

else d_auto=0;

if p_moto=1 then d_moto=1;

else d_moto=0;

if p_scoot=1 then d_scoot=1;

else d_scoot=0;

if p_telfix=1 then d_telfix=1;

else d_telfix=0;

if p_telcel=1 then d_telcel=1;

else d_telcel=0;

if p_tv=1 then d_tv=1;

else d_tv=0;

if p_vcr=1 then d_vcr=1;

else d_vcr=0;

if p_hifi=1 then d_hifi=1;

else d_hifi=0;

if p_pc=1 then d_pc=1;

else d_pc=0;run;
L’idea è quella di implementare un modello di regressione logistica (si veda nel riquadro il programma per il MODELLO 1) che ha come variabile risposta la dummy topred, e come variabili esplicative tutte le altre.

Si noti che l’opzione utilizzata event=’1’ consente che la PROC LOGISTIC modelli la probabilità che topred=1, in caso contrario la procedura modella la probabilità per l’evento con codice più basso (nel nostro caso il modello sarebbe stato implementato di default per topred=0).

***********MODELLO 1*************;
proc logistic;

model topred(event='1')=perc2 dir d_frigo--d_pc;

run;

Lettura dell’output

L’output di questa procedura risulta molto lungo, dato che include, oltre alle informazioni sulle stime dei parametri, una lunga serie di indici sulla bontà di adattamento e sui test applicati.

Seguono Output e istruzioni per una lettura rapida.

                                    La procedura LOGISTIC

                                  Informazioni sul modello

                     Data set                          WORK.PROVA

                     Variabile di risposta             topred

                     Numero di livelli di risposta     2

                     Modello                           logit binario

                     Tecnica di ottimizzazione         Scoring di Fisher

                           Numero di osservazioni lett       23918

                           Numero di osservazioni usat       23918

                                     Profilo di risposta

                               Valore                  Frequenza

                             ordinato       topred        totale

                                    1            0         18869

                                    2            1          5049

                           La probabilità del modello è topred=1.

                             Stato della convergenza del modello

                     Criterio di convergenza (GCONV=1E-8) soddisfatto.

                           Statistiche di adattamento del modello

                                                       Intercetta

                                              Solo              e

                           Criterio     intercetta      covariate

                           AIC           24656.937      17017.394

                           SC            24665.020      17154.795

                           -2 LOG L      24654.937      16983.394

                          Test dell'ipotesti nulla globale: BETA=0

                Test                 Chi-quadrato       DF     Pr > ChiQuadr

                Rapp. verosim.          7671.5432       16            <.0001

                Score                   6682.2820       16            <.0001

                Wald                    4100.7865       16            <.0001

                                    La procedura LOGISTIC

                       Analisi delle stime di massima verosimiglianza

                                           Errore    Chi-quadrato

          Parametro    DF       Stima    standard            Wald    Pr > ChiQuadr

          Intercept     1     -5.5587      0.3691        226.7766           <.0001

          perc2         1      2.1191      0.0497       1816.8958           <.0001

          dir           1      1.4657      0.0756        375.3704           <.0001

          d_frigo       1     -0.0801      0.3046          0.0692           0.7925

          d_lavast      1      0.6410      0.0402        254.3735           <.0001

          d_lavatr      1     -0.0437      0.2116          0.0426           0.8364

          d_ap_pul      1      0.3748      0.0685         29.9320           <.0001

          d_arcond      1      0.4768      0.0553         74.3707           <.0001

          d_auto        1      0.5067      0.1053         23.1578           <.0001

          d_moto        1      0.3829      0.0629         37.0110           <.0001

          d_scoot       1      0.0917      0.0437          4.4021           0.0359

          d_telfix      1      0.4966      0.1006         24.3582           <.0001

          d_telcel      1      0.7209      0.0695        107.5116           <.0001

          d_tv          1     -0.2491      0.1856          1.8020           0.1795

          d_vcr         1      0.2681      0.0662         16.3967           <.0001

          d_hifi        1      0.5094      0.0490        108.0015           <.0001

          d_pc          1      0.6527      0.0412        250.6019           <.0001

                                Stime dei rapporti di quote

                                    Stima    95% - Limiti di confidenza

                     Effetto     puntuale              di Wald

                     perc2          8.324       7.551             9.176

                     dir            4.330       3.734             5.023

                     d_frigo        0.923       0.508             1.677

                     d_lavast       1.898       1.755             2.054

                     d_lavatr       0.957       0.632             1.449

                     d_ap_pul       1.455       1.272             1.664

                     d_arcond       1.611       1.445             1.795

                     d_auto         1.660       1.350             2.040

                     d_moto         1.467       1.296             1.659

                     d_scoot        1.096       1.006             1.194

                     d_telfix       1.643       1.349             2.001

                     d_telcel       2.056       1.794             2.356

                     d_tv           0.780       0.542             1.121

                     d_vcr          1.307       1.148             1.489

                     d_hifi         1.664       1.512             1.832

                     d_pc           1.921       1.772             2.082

                  Associazione di probabilità previste e risposte osservate

                 Percentuale concordante        86.1    D di Somers    0.731

                 Percentuale discordante        13.0    Gamma          0.738

                 Percentuale legato              0.9    Tau-a          0.243

                 Coppie                     95269581    c              0.865 

In generale per una lettura veloce si consiglia di controllare:

- la colonna Pr>Chi-Quadr in corrispondenza delle stime dei parametri nella parte “Analisi delle stime di massima verosimiglianza”. Dato che per ogni parametro viene testata l’ipotesi nulla che esso sia uguale a 0 (ossia che non sussista un legame tra la relativa variabile esplicativa e la variabile risposta) si ritengono significativi i parametri con valore Pr>Chi-Quadr inferiore a 0,05, per i quali cioè viene respinta l’ipotesi nulla;

- le statistiche AIC e SC nella parte “Statistiche di adattamento del modello” utili per confrontare modelli diversi (valori bassi identificano il modello migliore);

- nella parte finale, “Associazione di probabilità previste e risposte osservate”, una percentuale alta di “percentuale concordante” indica una buona adattabilità del modello.

Esercizio 1

Si rifletta sul fatto che la variabile esplicativa perc2 è stata originata sulla base della variabile quantitativa discreta numperc, che in quanto quantitativa può essere utilizzata senza necessità di una codifica binaria. Si provi ora a modificare il MODELLO 1, sostituendo perc2 con numperc e si osservi come cambia l’output.

***********MODELLO 2*************;
proc logistic;

model topred(event='1')=numperc dir d_frigo--d_pc;

run;

Esercizio 2

Si costruisca un nuovo modello (MODELLO 1A) escludendo tutte le variabili che non risultano significative dal MODELLO 1. Si confronti l’output risultante con quello prodotto dal MODELLO 1.

Se ora si ipotizza una ulteriore relazione tra il benessere economico (in termini di reddito) e la grandezza dell’abitazione e il numero di componenti della famiglia. Per implementare un modello che consideri anche tale legame si utilizzeranno, insieme alle altre variabili esplicative anche le variabili quantitative: stanze e numcomp. Il modello è implementato secondo il seguente programma.

***********MODELLO 3*************;
proc logistic;

model topred(event='1')=numperc numcomp stanze dir d_lavast

d_ap_pul--d_moto d_telfix d_telcel d_vcr--d_pc;

run;
Dalla lettura dell’output si evince che delle due nuove variabili inserite nel modello, solo il numero di stanze risulta significativo; il modello risulta nel complesso migliorato.

Opzioni per la selezione delle variabili esplicative

Le operazioni  per selezionare le variabili esplicative significative, sulla base del risultato del test, possono essere programmate mediante l’utilizzo di alcune opzioni. In tal modo è il programma stesso a selezionare le variabili significative per il modello, mediante l’utilizzo di metodi differenti a seconda dell’opzione selezionata. Si suggerisce tuttavia di non affidarsi completamente a questi automatismi e di verificare sempre attraverso una procedura base l’effettiva significatività dei parametri.

Per scegliere un particolare metodo di selezione delle variabili esplicative, in modo che il modello finale inglobi solo quelle significative, si utilizza l’opzione selection dell’istruzione model.

Opzione stepwise
La procedura si compone di più passi, per ogni passo si considera il ruolo di ogni singola covariata e si valuta la bontà del modello; le variabili per quali il livello di significatività osservato è superiore a una certa soglia (lo 0,05 di default, usare l’opzione SLE= per indicare un valore soglia differente) vengono progressivamente eliminate. Si veda il programma nel riquadro seguente.

***********MODELLO con opzione STEPWISE*************;
proc logistic;

model topred(event='1')=numperc numcomp stanze dir

d_frigo--d_pc/selection=s;

run;
Opzione backward elimination
Il modello viene stimato in più passi: al passo 0 tutte le covariate vengono inglobate nel modello; ai passi successivi si eliminano progressivamente tutte le covariate per quali il livello di significatività osservato è superiore a una certa soglia (lo 0,05 di default, usare l’opzione SLE= per indicare un valore soglia differente). Si veda il programma nel riquadro seguente.

***********MODELLO con opzione BACKWARD ELIMINATION*************;
proc logistic;

model topred(event='1')=numperc numcomp stanze dir

d_frigo--d_pc/selection=b;

run;
ESEMPI:

data prova;set prova;

if titstu1 in (1,2,3) then titstu_d1=1;

else titstu_d1=0;

if titstu1 eq 4 then titstu_d2=1;

else titstu_d2=0;run;

PROC logistic;

model p_lavast(event='1')=numcomp;run;

PROC logistic;

model p_lavast(event='1')=numcomp sesso1 ;run;

PROC logistic;

model p_lavast(event='1')=numcomp sesso1 eta1;run;

PROC logistic;

model p_lavast(event='1')=numcomp sesso1 eta1 titstu_d1 titstu_d2;run;

PROC logistic;

model p_lavast(event='1')=numcomp sesso1 eta1 titstu_d1 titstu_d2 stanze;run;

PROC logistic;

model p_lavast(event='1')=numcomp sesso1 eta1 titstu_d1 titstu_d2 stanze p_arcond ;run;

PROC logistic;

model p_lavast(event='1')=numcomp sesso1 eta1 titstu_d1 titstu_d2 stanze p_arcond p_pc;run;

PROC logistic;

model p_lavast(event='1')=numcomp sesso1 eta1 titstu_d1 titstu_d2 stanze p_arcond p_pc p_hifi c_elettr c_acqua c_gas_r/selection=s;run;

